

## LOG IN

☐ Remember me

Create an account  
(<https://erlangcentral.org/login/?action=register>)

Login using:



([https://erlangcentral.org/login/?action=wordpress\\_social\\_authenticate&mode](https://erlangcentral.org/login/?action=wordpress_social_authenticate&mode))



([https://erlangcentral.org/login/?action=wordpress\\_social\\_authenticate&mode](https://erlangcentral.org/login/?action=wordpress_social_authenticate&mode))



([https://erlangcentral.org/login/?action=wordpress\\_social\\_authenticate&mode](https://erlangcentral.org/login/?action=wordpress_social_authenticate&mode))



([https://erlangcentral.org/login/?action=wordpress\\_social\\_authenticate&mode](https://erlangcentral.org/login/?action=wordpress_social_authenticate&mode))

[HOME \(HTTPS://ERLANGCENTRAL.ORG/\)](https://erlangcentral.org/) [NEWS](#) [JOBS](#) [RESOURCES](#) [LEARN ERLANG](#) [EVENTS](#) [USERS](#)

[FORUM \(HTTPS://ERLANGCENTRAL.ORG/FORUMS\)](https://erlangcentral.org/forums)

SEARCH:

## A Fast Web Server Demonstrating Some Undocumented Erlang Features

From ErlangCentral Wiki

### Contents

- 1 Author
- 2 Overview
- 3 TCP Server Framework
- 4 Common header file
- 5 Listening Process
- 6 Acceptor/Socket process
- 7 Web server state machine
- 8 Setting up the web server
- 9 Building a web application
- 10 Supervisor and Application implementation
- 11 License
- 12 Disclaimer

### Author

Sean

### Overview

This HOWTO describes a web server written for the day when even Yaws (<https://erlangcentral.org/frame/?href=http%3A%2F%2Fyaws.hyber.org>) is not quick enough.

The web server presented is quite simple. Even so it is split into 5 modules. Some of these are dictated by the OTP framework, and others are split out for convenience. The 5 modules are:

iserve - API for managing URIs and callbacks  
iserve\_app - OTP Application behaviour  
iserve\_sup - OTP Supervisor  
iserve\_server - Gen\_server to own the listening socket and create connections  
iserve\_socket - Process to handle a single HTTP connection for its lifetime

This HOWTO presents code and descriptions for each of these as they arise.

### TCP Server Framework

A web server needs to support lots of connections, so at it's heart it needs to be a multiple connection TCP/IP server. There are any number of ways to arrange a set of erlang processes into such a thing. My favourite method is to have a single gen\_server which opens and owns the listen socket (the listening process). This spawns another process which waits in accept until a connection attempt is received. At this time this accepting process sends a message back to the listening process and goes on to handle the traffic. This avoids the need for gen\_tcp:controlling\_process/2 and associated complexity.

On receipt of the message from the accepting process the listening process spawns a new accepting process and so on.

The listening process also traps exits, and if it receives a non normal exit from the current accepting process it creates a new one. In this way the listening process supervises its acceptor.

### Common Header File

The web server creates a #req{} record as it processes each request. This is used as part of the API into implementation callbacks and by the iserve\_socket process. Here are the contents of iserve.hrl up front to get it out of the way:

```
1 % This record characterises the connection from the browser to our server ?
2 % it is intended to be a consistent view derived from a bunch of different headers
3 -record(req, {connection=keep_alive, % keep_alive | close
4             content_length, % Integer
5             vsn, % {Maj,Min}
6             method, % 'GET' | 'POST'
7             uri, % Truncated URI /index.html
8             args="", % Part of URI after ?
9             headers, % [{Tag, Val}]
10            body = <<>>}). % Content Body
```

### Listening Process

Here is the code for the listening process. It is a very basic gen\_server which models a single process:

```
1 -module(iserve_server). ?
2
3 -behaviour(gen_server).
4
5 -export([start_link/1, create/2]).
6
7 %% gen_server callbacks
8 -export([init/1, handle_call/3, handle_cast/2, handle_info/2, terminate/2,
```

```

9         code_change/3]).
10
11 -record(state, {listen_socket,
12                 port,
13                 acceptor}).
14
15 %%-----
16 start_link(Port) when is_integer(Port) ->
17     Name = list_to_atom(lists:flatten(io_lib:format("iserve_~w", [Port]))),
18     gen_server:start_link({local, Name}, ?MODULE, Port, []).
19
20 %% Send message to cause a new acceptor to be created
21 create(ServerPid, Pid) ->
22     gen_server:cast(ServerPid, {create, Pid}).
23
24
25 %% Called by gen_server framework at process startup. Create listening socket
26 init(Port) ->
27     process_flag(trap_exit, true),
28     case gen_tcp:listen(Port,[binary, {packet, http},
29                             {reuseaddr, true},
30                             {active, false},
31                             {backlog, 30}]) of
32     {ok, Listen_socket} ->
33         %%Create first accepting process
34         Pid = iserve_socket:start_link(self(), Listen_socket, Port),
35         {ok, #state{listen_socket = Listen_socket,
36                     port = Port,
37                     acceptor = Pid}};
38     {error, Reason} ->
39         {stop, Reason}
40     end.
41
42
43 handle_call(_Request, _From, State) ->
44     Reply = ok,
45     {reply, Reply, State}.
46
47 %% Called by gen_server framework when the cast message from create/2 is received
48 handle_cast({create, _Pid}, #state{listen_socket = Listen_socket} = State) ->
49     New_pid = iserve_socket:start_link(self(), Listen_socket, State#state.port),
50     {noreply, State#state{acceptor=New_pid}};
51
52 handle_cast(_Msg, State) ->
53     {noreply, State}.
54
55
56 handle_info({'EXIT', Pid, normal}, #state{acceptor=Pid} = State) ->
57     {noreply, State};
58
59 %% The current acceptor has died, wait a little and try again
60 handle_info({'EXIT', Pid, _Abnormal}, #state{acceptor=Pid} = State) ->
61     timer:sleep(2000),
62     iserve_socket:start_link(self(), State#state.listen_socket, State#state.port),
63     {noreply, State};
64
65 handle_info(_Info, State) ->
66     {noreply, State}.
67
68
69 terminate(Reason, State) ->
70     gen_tcp:close(State#state.listen_socket),
71     ok.
72
73
74 code_change(_OldVsn, State, _Extra) ->
75     {ok, State}.

```

The notable thing about this code is the use of undocumented socket options to set up the initial state of connections made to the web server port.

{backlog, 30} specifies the length of the OS accept queue.

{packet, http} puts the socket into http mode. This makes the socket wait for a HTTP Request line, and if this is received to immediately switch to receiving HTTP header lines. The socket stays in header mode until the end of header marker is received (CR,NL,CR,NL), at which time it goes back to wait for a following HTTP Request line.

## Acceptor/Socket Process

It would be easy enough to create an abstraction of the Listen/Accept process structure and pass in the implementation function as another parameter. For this HOWTO however I'll stick with the most basic model - the acceptor process starts life as an acceptor and goes on to handle the traffic.

The acceptor process is implemented in a separate module iserve\_socket. It is in two parts - the first part sets up a bunch of defines and exports and then does the accepting. Here it is:

```

1  -module(iserve_socket).
2
3  -export([start_link/3]).
4
5  -export([init/1]).
6  -include("iserve.hrl").
7
8  -define(not_implemented_501, "HTTP/1.1 501 Not Implemented\r\n\r\n").
9  -define(forbidden_403, "HTTP/1.1 403 Forbidden\r\n\r\n").
10 -define(not_found_404, "HTTP/1.1 404 Not Found\r\n\r\n").
11
12 -record(c, {sock,
13             port,
14             peer_addr,
15             peer_port
16             }).
17

```

```

18 -define(server_idle_timeout, 30*1000).
19
20 start_link(ListenPid, ListenSocket, ListenPort) ->
21   proc_lib:spawn_link(?MODULE, init, [{ListenPid, ListenSocket, ListenPort}]).
22
23 init({Listen_pid, Listen_socket, ListenPort}) ->
24   case catch gen_tcp:accept(Listen_socket) of
25     {ok, Socket} ->
26       %% Send the cast message to the listener process to create a new acceptor
27       iserve_server:create(Listen_pid, self()),
28       {ok, {Addr, Port}} = inet:peername(Socket),
29       C = #c{sock = Socket,
30               port = ListenPort,
31               peer_addr = Addr,
32               peer_port = Port},
33       request(C, #req{}); %% Jump to state 'request'
34     Else ->
35       error_logger:error_report([application, iserve],
36                                "Accept failed error",
37                                io_lib:format("~p", [Else])),
38     exit({error, accept_failed})
39   end.

```

Note here that the process is started via the `proc_lib:spawn_link/3` call. This wraps the normal `spawn_link/3` bif so that the same nice error reports are created as for `gen_servers`, but it allows for a totally unstructured process implementation.

## Web Server State Machine

The rest of this module contains the web server code. It is structured as a state machine which follows the state changes of the http socket mode. A single function models each state, and state transitions are simply implemented as a call to the function which owns the next state.

The states are:

request - wait for a HTTP Request line. Transition to state headers if one is received.

headers - collect HTTP headers. After the end of header marker transition to body state.

body - collect the body of the HTTP request if there is one, and lookup and call the implementation callback. Depending on whether the request is persistent transition back to state request to await the next request or exit.

The code for the state request is below. A blocking call is made to `gen_tcp:recv/3` with a timeout. The http driver waits for a CRNL terminated line of the form GET / HTTP/1.0. If anything else is received an `http_error` indication is returned with the erroneous data.

Some broken clients include extra CR or CRNL sequences so these are skipped.

```

1 request(C, Req) ->
2   case gen_tcp:recv(C#c.sock, 0, 30000) of
3     {ok, {http_request, Method, Path, Version}} ->
4       headers(C, Req#req{vsn = Version,
5                          method = Method,
6                          uri = Path}, []);
7     {error, {http_error, "\r\n"}} ->
8       request(C, Req);
9     {error, {http_error, "\n"}} ->
10      request(C, Req);
11     _Other ->
12       exit(normal)
13   end.

```

The code for the state headers is below. After sending the HTTP request line the http driver automatically switches into header receive mode. The driver looks for values of the form Header-Val: value and sends them one by one after each call to `recv`.

The driver maintains a hash table of well known header values and if one of those is received from the network it returns the header value as an atom. Otherwise the header value is returned as a string. In both cases the driver takes care of case insensitivity and automatically capitalises the first letter of each hyphen separated word in the header name. The author clearly got a little carried away at this point!

This web server extracts the values of the 'Content-Length' and 'Connection' headers for its own purposes and simply accumulates the other headers in a list to be passed to the application callback.

Erlang Central Cafe ▲

At the end of the headers the driver returns `{ok, http_eoh}`. This is the cue for the web server to skip to body mode. The driver automatically switches to wait for a new request line at this point unless a subsequent call to `inet:setopts/2` is made.

```

1 headers(C, Req, H) ->
2   case gen_tcp:recv(C#c.sock, 0, ?server_idle_timeout) of
3     {ok, {http_header, _, 'Content-Length', _, Val}} ->
4       Len = list_to_integer(Val),
5       headers(C, Req#req{content_length = Len}, [{ 'Content-Length', Len }|H]);
6     {ok, {http_header, _, 'Connection', _, Val}} ->
7       Keep_alive = keep_alive(Req#req.vsn, Val),
8       headers(C, Req#req{connection = Keep_alive}, [{ 'Connection', Val }|H]);
9     {ok, {http_header, _, Header, _, Val}} ->
10      headers(C, Req, [{Header, Val}|H]);
11     {error, {http_error, "\r\n"}} ->
12      headers(C, Req, H);
13     {error, {http_error, "\n"}} ->
14      headers(C, Req, H);
15     {ok, http_eoh} ->
16      body(C, Req#req{headers = lists:reverse(H)});
17     _Other ->
18       exit(normal)
19   end.
20
21 %% Shall we keep the connection alive?
22 %% Default case for HTTP/1.1 is yes, default for HTTP/1.0 is no.
23 %% Exercise for the reader - finish this so it does case insensitivity properly !
24 keep_alive({1,1}, "close") -> close;

```

```

25 keep_alive({1,1}, "Close") -> close;
26 keep_alive({1,1}, _) -> keep_alive;
27 keep_alive({1,0}, "Keep-Alive") -> keep_alive;
28 keep_alive({1,0}, _) -> close;
29 keep_alive({0,9}, _) -> close;
30 keep_alive(Vsn, KA) ->
31   io:format("Got = ~p~n",[{Vsn, KA}]),
32   close.

```

The code for the state body is below. At this point we have everything required except the body in the case of a POST request. If present this is retrieved in a single chunk based on the content length supplied. Most web servers will implement some sort of size limit for POST requests. This is still needed in our case to avoid a single client taking all the memory of the Erlang Virtual machine with the subsequent crash. It should be simple to add.

Unless the connection is a keep-alive type the process terminates at the end of processing this function. All resources are cleared up at process exit including open sockets so we do not need to be too careful about explicitly tidying up.

```

1  body(#c{sock = Sock} = C, Req) ->
2    case Req#req.method of
3      'GET' ->
4        Close = handle_get(C, Req),
5        case Close of
6          close ->
7            gen_tcp:close(Sock);
8            keep_alive ->
9              inet:setopts(Sock, [{packet, http}]),
10             request(C, #req{})
11          _;
12        'POST' when is_integer(Req#req.content_length) ->
13          inet:setopts(Sock, [{packet, raw}]),
14          case gen_tcp:recv(Sock, Req#req.content_length, 60000) of
15            {ok, Bin} ->
16              Close = handle_post(C, Req#req{body = Bin}),
17              case Close of
18                close ->
19                  gen_tcp:close(Sock);
20                  keep_alive ->
21                    inet:setopts(Sock, [{packet, http}]),
22                    request(C, #req{})
23                _;
24              _Other ->
25                exit(normal)
26            end;
27          _Other ->
28            send(C, ?not_implemented_501),
29            exit(normal)
30        end.

```

The rest of the `iserve_socket` module is below. There is not much left to do. The `inet` driver has already worked out for us what sort of URI is being used.

The `call_mfa/4` function relies on the existence of an `ets/mnesia` table which converts the URI into a module and function dynamic callback. This must have been created at installation (see section later).

```

1  handle_get(C, #req{connection = Conn} = Req) ->
2    case Req#req.uri of
3      {abs_path, Path} ->
4        {F, Args} = split_at_q_mark(Path, []),
5        call_mfa(F, Args, C, Req),
6        Conn;
7      {absoluteURI, http, _Host, _, Path} ->
8        {F, Args} = split_at_q_mark(Path, []),
9        call_mfa(F, Args, C, Req),
10       Conn;
11      {absoluteURI, _Other_method, _Host, _, _Path} ->
12        send(C, ?not_implemented_501),
13        close;
14      {scheme, _Scheme, _RequestString} ->
15        send(C, ?not_implemented_501),
16        close;
17      _ ->
18        send(C, ?forbidden_403),
19        close
20    end.
21
22  handle_post(C, #req{connection = Conn} = Req) ->
23    case Req#req.uri of
24      {abs_path, Path} ->
25        call_mfa(Path, Req#req.body, C, Req),
26        Conn;
27      {absoluteURI, http, _Host, _, Path} ->
28        call_mfa(Path, Req#req.body, C, Req),
29        Conn;
30      {absoluteURI, _Other_method, _Host, _, _Path} ->
31        send(C, ?not_implemented_501),
32        close;
33      {scheme, _Scheme, _RequestString} ->
34        send(C, ?not_implemented_501),
35        close;
36      _ ->
37        send(C, ?forbidden_403),
38        close
39    end.
40
41  call_mfa(F, A, C, Req) ->
42    case iserve:lookup(C#c.port, Req#req.method, F) of
43      {ok, Mod, Func} ->
44        case catch Mod:Func(Req, A) of
45          {'EXIT', Reason} ->

```

```

46         io:format("Worker Crash = ~p~n",[Reason]),
47         exit(normal);
48     {200, Headers0, Body} ->
49         Headers = add_content_length(Headers0, Body),
50         Enc_headers = enc_headers(Headers),
51         Resp = [<<"HTTP/1.1 200 OK\r\n">>,
52                 Enc_headers,
53                 <<"\r\n">>,
54                 Body],
55         send(C, Resp)
56     end;
57     {error, not_found} ->
58         send(C, ?not_found_404)
59     end.
60
61 add_content_length(Headers, Body) ->
62     case lists:keysearch('Content-Length', 1, Headers) of
63     {value, _} ->
64         Headers;
65     false ->
66         [{'Content-Length', size(Body)}|Headers]
67     end.
68
69
70 enc_headers([{Tag, Val}|T]) when is_atom(Tag) ->
71     [atom_to_list(Tag), ": ", enc_header_val(Val), "\r\n"|enc_headers(T)];
72 enc_headers([{Tag, Val}|T]) when is_list(Tag) ->
73     [Tag, ": ", enc_header_val(Val), "\r\n"|enc_headers(T)];
74 enc_headers([]) ->
75     [].
76
77 enc_header_val(Val) when is_atom(Val) ->
78     atom_to_list(Val);
79 enc_header_val(Val) when is_integer(Val) ->
80     integer_to_list(Val);
81 enc_header_val(Val) ->
82     Val.
83
84 %% Split the path at the ?. This would have to do all sorts of
85 %% horrible ../ path checks and %C3 etc decoding if we wanted to
86 %% retrieve actual paths to real filesystem files. As it is we only
87 %% want to look it up as a key in mnesia/ets :)
88 split_at_q_mark([$_|T], Acc) ->
89     {lists:reverse(Acc), T};
90 split_at_q_mark([H|T], Acc) ->
91     split_at_q_mark(T, [H|Acc]);
92 split_at_q_mark([], Acc) ->
93     {lists:reverse(Acc), []}.
94
95
96 send(#c{sock = Sock}, Data) ->
97     case gen_tcp:send(Sock, Data) of
98     ok ->
99         ok;
100     _ ->
101         exit(normal)
102     end.

```

## Setting Up The Web Server

The Web server requires two preparation steps. The port number the web server listens on is defined in a file called `iserve.conf` which must be located in the `priv` subdirectory of the `iserve` application. It must contain a line of the form:

```
{port, 8081}.
```

If this file is not present then the port number defaults to 8080.

The web server also uses an `mnesia` table to manage mappings between URLs and implementation callbacks. This may be created and managed with the `iserve.erl` module:

```

1  -module(iserve).
2  -export([create_table/1,
3           add_callback/5, delete_callback/3,
4           print_callbacks/0,lookup/3]).
5
6  -record(iserve_callback, {key,           % {Port, 'GET'|'POST', Abs_path}
7                           mf}).         % {Mod, Func}
8
9  create_table(Nodes) ->
10     mnesia:create_table(iserve_callback,
11                          [{attributes, record_info(fields, iserve_callback)},
12                           {disc_copies, Nodes}]).
13
14 lookup(Port, Method, Path) ->
15     case ets:lookup(iserve_callback, {Port, Method, Path}) of
16     [#iserve_callback{mf = {Mod, Func}}] ->
17         {ok, Mod, Func};
18     [] ->
19         {error, not_found}
20     end.
21
22 add_callback(Port, Method, Path, Mod, Func) when ((Method == 'GET') or (Method == 'POST')) and
23                                                    is_list(Path) and is_atom(Mod) and
24                                                    is_atom(Func) and is_integer(Port) ->
25     mnesia:dirty_write(iserve_callback, #iserve_callback{key = {Port, Method, Path},
26                                                            mf = {Mod, Func}}).
27
28
29 delete_callback(Port, Method, Path) ->
30     mnesia:dirty_delete(iserve_callback, {Port, Method, Path}).
31

```

```

32 | print_callbacks() ->
33 |     All = mnesia:dirty_match_object(#iserve_callback{ _ = '_' },
34 |     io:format("Port\tMethod\tPath\tModule\tFunction~n"),
35 |     lists:foreach(fun(#iserve_callback{key = {Port, Method, Path},
36 |                       mf = {Module, Function}}) ->
37 |                     io:format("~p\t~p\t~p\t~p\t~p\t~p~n", [Port, Method, Path, Module, Functi
38 |                     end, All)).

```

iserve:create\_table([node()]). must be called once at installation.

All Urls must be stored in this table with a module and function which will create the page. So for example the callback for the document root might be defined with:

iserve:add\_callback(8081, 'GET', "/", test\_iserve\_app, do\_get).

The callback for index.html could be:

iserve:add\_callback(8081, 'GET', "/index.html", module, function2).

## Building A Web Application

The simplest kind of iserve web application would be one to simply return a generated page. A function must be implemented which returns {200, Headers, Body} where Headers is a list of {Header-Atom, Val-String} and Body is a binary. For example:

```

1 | -module(test_iserve_app).
2 | -export([do_get/2]).
3 | -include("iserve.hrl").
4 |
5 | do_get(#req{ } = Req, Args) ->
6 |     {200, [], <<"<!DOCTYPE html PUBLIC "-//W3C//DTD HTML 4.01 Transitional//EN">
7 |     <html>
8 |     <head>
9 |     <title>Welcome to iserve</title>
10 |    </head>
11 |    <body>
12 |    Hello
13 |    </body>
14 |    </html>>>}.

```

Obviously this is an extremely simple example. This is where you come in!

## Supervisor And Application Implementation

The web server only needs a little help to become a full blown OTP application. It needs an application behaviour, a supervisor behaviour, and a .app file.

These are presented below.

The Application:

```

1 | -module(iserve_app).
2 | -behaviour(application).
3 | -export([
4 |     start/2,
5 |     stop/1
6 | ]).
7 |
8 | start(_Type, _StartArgs) ->
9 |     case iserve_sup:start_link() of
10 |     {ok, Pid} ->
11 |         alarm_handler:clear_alarm({application_stopped, iserve}),
12 |         {ok, Pid};
13 |     Error ->
14 |         alarm_handler:set_alarm({application_stopped, iserve, []}),
15 |         Error
16 |     end.
17 |
18 | stop(_State) ->
19 |     alarm_handler:set_alarm({application_stopped, iserve, []}),
20 |     ok.

```

The Supervisor:

```

1 | -module(iserve_sup).
2 | -behaviour(supervisor).
3 | -export([
4 |     start_link/0,
5 |     init/1
6 | ]).
7 |
8 | -define(SERVER, ?MODULE).
9 |
10 | start_link() ->
11 |     supervisor:start_link({local, ?SERVER}, ?MODULE, []).
12 |
13 | init([]) ->
14 |     Port = get_config(),
15 |     Server = {iserve_server, {iserve_server, start_link, [Port]},
16 |               permanent, 2000, worker, [iserve_server]},
17 |     {ok, {{one_for_one, 10, 1}, [Server]}}.
18 |
19 | get_config() ->
20 |     case file:consult(filename:join(code:priv_dir(iserve), "iserve.conf")) of
21 |     [{port, Port}] ->
22 |         Port;
23 |     _ ->
24 |         8080
25 |     end.

```

The .app file.

A dependency on sasl is only included because of the calls to set and clear alarms in the application behaviour implementation:

```
1 {application, iserve,  
2   [{description, "Web Server"},  
3     {vsn, "%ISERVE_VSN%"},  
4     {modules, [ iserve_sup,  
5                 iserve_app,  
6                 iserve_server,  
7                 iserve_socket  
8               ]},  
9  
10    {registered, [ iserve_sup]},  
11    {applications, [kernel, stdlib, sasl]},  
12    {mod, {iserve_app, []}}]}.  
?
```

## License

The code associated with this HOWTO is available under the BSD License (<https://erlangcentral.org/frame/?href=http%3A%2F%2Fwww.opensource.org%2Flicenses%2Fbsd-license.php>)

## Disclaimer

The undocumented features presented in this HOWTO are undocumented because they are not supported by Ericsson. On the other hand they are used in commercially shipping systems.

Retrieved from 'http://erlangcentral.org/wiki/index.php?

title=A\_fast\_web\_server\_demonstrating\_some\_undocumented\_Erlang\_features&oldid=37315

(http://erlangcentral.org/wiki/index.php?

title=A\_fast\_web\_server\_demonstrating\_some\_undocumented\_Erlang\_features&oldid=37315'

### Our Sponsors:

(<https://erlangcentral.org/industrial-erlang-user-group/member-companies/>)

**GOFACTORY** (<http://go-factory.com/>)

### Follow us:



(<http://www.facebook.com/pages/Erlang-Central/174661552676546>)



(<https://twitter.com/ErlangCentral>)



(<https://erlangcentral.org/feed/rss/>)

Terms of use and privacy policy

(<https://erlangcentral.org/terms-of-use-privacy-policy/>)

About Us (<https://erlangcentral.org/contact/about-us/>) |

Contact Us (<https://erlangcentral.org/contact/contact-us/>) |

FAQ (<https://erlangcentral.org/contact/faq/>) | Roadmap

(<https://erlangcentral.org/contact/erlang-central-roadmap/>)

| Cafe Help (<https://erlangcentral.org/cafe-help/>)

Industrial Erlang User Group

(<https://erlangcentral.org/industrial-erlang-user-group/>)